Lesson 16: Integration

Definite integrals

When Maple can find an antiderivative of a function, it should have no trouble with definite integrals of that function, using the Fundamental Theorem of Calculus. Well, maybe...

Here's an example of what can go wrong.

```
> f:= abs(sin(x))^3;
```

$$f := |\sin(x)|^3$$

```
> F:= unapply(int(f,x),x);
```

$$F := x \to \frac{1}{3}\,\mathrm{signum}(\sin(x))^3 \cos(x)\,\left(-3+\cos(x)^2\right)$$

Note $\mathrm{signum}(t) = 1$ for $t > 0$, -1 for $t < 0$.

It looks OK as an antiderivative:

```
> plot(D(F)(x)-f,x=0..2*Pi);
```



So let's try a definite integral.

```
> int(f,x=Pi/2 .. 3*Pi/2) = F(3*Pi/2) - F(Pi/2);
```

$$\frac{4}{3} = 0$$

Which is right?

Well, f is $\sin(x)^3$ for $0 < x < \pi$ and $-\sin(x)^3$ for $\pi < x < 2\pi$.

```
> int(sin(x)^3,x=Pi/2 .. Pi) - int(sin(x)^3,x=Pi .. 3*Pi/2);
```

$$\frac{4}{3}$$

If we plot the antiderivative, we see what's going on.

```
> plot(F(x),x=0..2*Pi);
```



So Maple was right.
The point of this story: you also have to be careful about jumps in the "antiderivative".
Maple doesn't always succeed in locating them, though it does this time. It's better than it used to be, but still not perfect. Absolute values are especially troublesome.
In this one, Maple doesn't give us a value for the integral.

```
> int(abs(x-cos(x)),x=-Pi..Pi);
```

$$\int_{-\pi}^{\pi} |x-\cos(x)|\,dx$$

```
> int(abs(x-cos(x)),x);
```

$$-\frac{1}{2}\operatorname{signum}(x-\cos(x))\left(-x^2+2\sin(x)\right)$$

```
> plot(%,x=-Pi..Pi);
```



The antiderivative takes a jump at the point where $x=\cos(x)$. But there's no exact formula for that, unless you use RootOf.

```
> solve(x=cos(x));
```

$$\mathit{RootOf}(\_Z-\cos(\_Z)) \qquad \textbf{(1.1)}$$

So in this case perhaps you can't blame Maple for not coming up with a result. But sometimes it's too cautious.

```
> int(1/(1+x^2 + x^5),x=0..1);
```

$$\int_{0}^{1} \frac{1}{1+x^2+x^5}\,dx$$

```
> F:= unapply(int(1/(1+x^2+x^5),x), x);
```

$$F := x \to \sum_{\_R = RootOf(3233\_Z^5 + 27\_Z^3 + 50\_Z^2 - 1)} \_R \ln\left(x - \frac{2224698426}{7363759}\_R^4 + \frac{797015325}{7363759}\_R^3 - \frac{109507419}{7363759}\_R^2 + \frac{18837605}{7363759}\_R + \frac{7092000}{7363759}\right)$$

```
> plot(F(x),x=0..1);
```



It looks continuous. Here's what the Fundamental Theorem of Calculus gives us.

```
> evalf(F(1) - F(0));
```

$$0.7398318370 + 0.\ I$$

Note that last bit is 0 i, not 0.1. Presumably the calculation involved complex numbers, and the imaginary parts cancelled. To get rid of the 0 i:

```
> simplify(%);
```

$$0.7398318370 \tag{1.2}$$

To check this, here's what numerical integration (which we'll talk about in detail later) gives us for this integral.

```
> evalf(Int(1/(1+x^2+x^5),x=0..1));
```

$$0.7398318370$$

Note the difference between **evalf(Int(...))** and **evalf(int(...))**: **evalf(Int(...))** will always do numerical integration without trying symbolic methods (because the **Int** is inert), **evalf(int(...))** tries symbolic integration first and applies **evalf** to whatever the result is.

If you're sure the antiderivative will be continuous (or you feel like living dangerously) you can tell Maple that:

```
> int(1/(1+x^2+x^5),x=0..1, continuous);
```

$$\sum_{\_R=RootOf(3233\_Z^5+27\_Z^3+50\_Z^2-1)} \_R\ln\left(-\frac{109507419}{7363759}\_R^2+\frac{14455759}{7363759}-\frac{2224698426}{7363759}\_R^4+\frac{797015325}{7363759}\_R^3+\frac{18837605}{7363759}\_R\right)-\left(\sum_{\_R=RootOf(3233\_Z^5+27\_Z^3+50\_Z^2-1)} \_R\ln\left(-\frac{2224698426}{7363759}\_R^4+\frac{797015325}{7363759}\_R^3-\frac{109507419}{7363759}\_R^2+\frac{18837605}{7363759}\_R+\frac{7092000}{7363759}\right)\right) \quad (1.3)$$

```
> evalf(%);
```

$$0.7398318370+0.\,\mathrm{I} \quad (1.4)$$

Here's a bug.

```
> J:= Int(arctan(z)^2/(1+4*z^2),z=-1..1);
```

$$J:=\int_{-1}^{1}\frac{\arctan(z)^2}{1+4z^2}\,dz \quad (1.5)$$

```
> V:= value(J);
```

$$V:=-\frac{1}{64}\mathrm{I}\pi^2\ln\left(1+\frac{1}{3}\mathrm{I}\right)+\frac{1}{16}\pi\,\mathrm{polylog}\left(2,-\frac{1}{3}\mathrm{I}\right)-\frac{1}{8}\mathrm{I}\,\mathrm{polylog}\left(3,-\frac{1}{3}\mathrm{I}\right)+\frac{1}{64}\mathrm{I}\pi^2\ln(1+3\mathrm{I})-\frac{1}{16}\pi\,\mathrm{polylog}(2,-3\mathrm{I})+\frac{1}{8}\mathrm{I}\,\mathrm{polylog}(3,-3\mathrm{I})+\frac{1}{64}\mathrm{I}\pi^2\ln\left(1-\frac{1}{3}\mathrm{I}\right)+\frac{1}{16}\pi\,\mathrm{polylog}\left(2,\frac{1}{3}\mathrm{I}\right)+\frac{1}{8}\mathrm{I}\,\mathrm{polylog}\left(3,\frac{1}{3}\mathrm{I}\right)-\frac{1}{64}\mathrm{I}\pi^2\ln(1-3\mathrm{I})-\frac{1}{16}\pi\,\mathrm{polylog}(2,3\mathrm{I})-\frac{1}{8}\mathrm{I}\,\mathrm{polylog}(3,3\mathrm{I}) \quad (1.6)$$

```
> evalf(V=J);
```

$$0.6466952900-2.\,10^{-11}\,\mathrm{I}=0.1727275411 \quad (1.7)$$

So there's a discrepancy. What's the antiderivative?

```
> F:= unapply(int(arctan(z)^2/(1+4*z^2),z),z);
```

$$F:=z\to\frac{1}{4}\mathrm{I}\arctan(z)^2\ln\left(1-\frac{1}{3}\frac{(1+\mathrm{I}z)^2}{z^2+1}\right)+\frac{1}{4}\arctan(z)\,\mathrm{polylog}\left(2,\frac{1}{3}\frac{(1+\mathrm{I}z)^2}{z^2+1}\right)+\frac{1}{8}\mathrm{I}\,\mathrm{polylog}\left(3,\frac{1}{3}\frac{(1+\mathrm{I}z)^2}{z^2+1}\right)-\frac{1}{4}\mathrm{I}\arctan(z)^2\ln\left(1-\frac{3(1+\mathrm{I}z)^2}{z^2+1}\right)-\frac{1}{4}\arctan(z)\,\mathrm{polylog}\left(2,\frac{3(1+\mathrm{I}z)^2}{z^2+1}\right)-\frac{1}{8}\mathrm{I}\,\mathrm{polylog}\left(3,\frac{3(1+\mathrm{I}z)^2}{z^2+1}\right) \quad (1.8)$$

Was Maple using the Fundamental Theorem of Calculus with this antiderivative?

```
> V - (F(1) - F(-1));
```

$$0 \tag{1.9}$$

Is it really an antiderivative?

```
> simplify(diff(F(z),z)-arctan(z)^2/(1+4*z^2));
```

$$0 \tag{1.10}$$

```
> plot(Re(F(z)),z=-1..1);
```



Improper integrals

Maple can do improper integrals. It checks for convergence or divergence. Here are some you may have met.

```
> int(1/x^2, x= 1 .. infinity);
```

$$1$$

```
> int(1/x, x= -1 .. 1);
```

$$undefined$$

```
> int(exp(-a*x),x=0..infinity);
```

$$\lim_{x \to \infty} \left(-\frac{e^{-a x} - 1}{a} \right)$$

Why didn't that one work? Because nobody told Maple that $a > 0$. If $a < 0$ the integral doesn't converge. Don't expect Maple to read your mind! And remember that variables are allowed to be complex unless otherwise specified, so replacing a with a^2 wouldn't have worked either. When you want to make an assumption on a variable, you can use **assume**.

```
> assume(a > 0);
```

```
> about(a);
Originally a, renamed a~:
  is assumed to be: RealRange(Open(0),infinity)
```

```
> int(exp(-a*x),x=0..infinity);
```

$$\frac{1}{a\sim}$$

That ~ is there to remind you that the variable has an assumption on it. You can change that: Tools, Options..., Display, and change Assumed Variables from Trailing Tildes to Phrase or No Annotation.

To remove assumptions from a variable, "unassign" it.

```
> a:= 'a';
```

$$a := a$$

```
> about(a);
a:
  nothing known about this object
```

```
> int(exp(-a*x),x=0..infinity);
```

$$\lim_{x \to \infty} \left(-\frac{e^{-a x} - 1}{a} \right)$$

You can also make a temporary assumption (for just a single command) using **assuming**.

```
> int(exp(-a*x),x=0..infinity) assuming a > 0;
```

$$\frac{1}{a}$$

```
> about(a);
a:
  nothing known about this object
```

Here's an improper integral that might appear in Math 301, which you would do using residues. Maple can do it too.

```
> J:= Int(x^a/(x^2+3*x+2),x=0..infinity);
  J = value(J) assuming a > 0, a < 1;
```

$$J := \int_0^{\infty} \frac{x^a}{x^2 + 3x + 2} \, dx$$

$$\int_0^{\infty} \frac{x^a}{x^2+3x+2}\,dx = 2^a\left(\pi\csc(\pi a) - \pi\csc(\pi a)\,2^{-a}\right)$$

Maple is not using an antiderivative on this one, by the way, because it doesn't know an antiderivative.

```
> int(x^a/(x^2 + 3*x + 2), x) assuming a > 0, a < 1;
```

$$\int \frac{x^a}{x^2+3x+2}\,dx$$

On the other hand, here's one that we can do easily with a bit of insight, but Maple can't.

```
> J:= Int(sin(a*x)/(1 + x^4 + x^6), x=-infinity .. infinity);
  value(J) assuming a > 0;
```

$$J := \int_{-\infty}^{\infty} \frac{\sin(a\,x)}{1+x^4+x^6}\,dx$$

$$\lim_{x\to-\infty}\left(-a^5\left(\sum_{\_R1=RootOf(a^6+\_Z^4a^2+\_Z^6)} \frac{1}{2}\,\frac{-\mathrm{Si}(-a\,x+\_R1)\cos(\_R1)+\mathrm{Ci}(a\,x-\_R1)\sin(\_R1)}{\_R1^3\,(2\,a^2+3\,\_R1^2)}\right) + a^5\left(\sum_{\_R1=RootOf(a^6+\_Z^4a^2+\_Z^6)} \frac{1}{4}\,\frac{\pi\cos(\_R1)}{\_R1^3\,(2\,a^2+3\,\_R1^2)}\right)\right)$$

... or even

```
> J1:=int(sin(a*x)/(1+x^4 + x^6), x=-1 .. 1);
```

$$J1 := -a^5\left(\sum_{\_R1=RootOf(a^6+\_Z^4a^2+\_Z^6)} \frac{1}{2}\,\frac{-\mathrm{Si}(a+\_R1)\cos(\_R1)+\mathrm{Ci}(-a-\_R1)\sin(\_R1)}{\_R1^3\,(2\,a^2+3\,\_R1^2)}\right) + a^5\left(\sum_{\_R1=RootOf(a^6+\_Z^4a^2+\_Z^6)} \frac{1}{2}\,\frac{-\mathrm{Si}(-a+\_R1)\cos(\_R1)+\mathrm{Ci}(a-\_R1)\sin(\_R1)}{\_R1^3\,(2\,a^2+3\,\_R1^2)}\right)$$

```
> simplify(%);
```

$$-\frac{1}{2}\,a^5\left(\left(\sum_{\_R1=RootOf(a^6+\_Z^4a^2+\_Z^6)} \frac{-\mathrm{Si}(a+\_R1)\cos(\_R1)+\mathrm{Ci}(-a-\_R1)\sin(\_R1)}{\_R1^3\,(2\,a^2+3\,\_R1^2)}\right) - \left(\sum_{\_R1=RootOf(a^6+\_Z^4a^2+\_Z^6)} \frac{-\mathrm{Si}(-a+\_R1)\cos(\_R1)+\mathrm{Ci}(a-\_R1)\sin(\_R1)}{\_R1^3\,(2\,a^2+3\,\_R1^2)}\right)\right)$$

Of course it can get a numerical value for particular values of a.

```
> (eval(J1,a=3)); evalf(%);
```

$$-243\left(\sum_{\_R1=RootOf(729+9\,\_Z^4+\_Z^6)} \frac{1}{2}\,\frac{-\mathrm{Si}(3+\_R1)\cos(\_R1)+\mathrm{Ci}(-3-\_R1)\sin(\_R1)}{\_R1^3\,(18+3\,\_R1^2)}\right)$$

$$+243\left(\sum_{\_R1=RootOf(729+9\_Z^4+\_Z^6)} \frac{1}{2}\frac{-\mathrm{Si}(-3+\_R1)\cos(\_R1)+\mathrm{Ci}(3-\_R1)\sin(\_R1)}{\_R1^3\,(18+3\,\_R1^2)}\right)$$

$$0.+0.\,\mathrm{I}$$

That's 0 i, not 0.1.

Exact or numeric?

In some cases, even when there is an "exact" formula for the antiderivative, that formula could be so complicated that numerical integration may be faster and more accurate. In general, **evalf(Int(...))** is quite reliable and reasonably fast, except for some circumstances we'll see where it has difficulties.

time() returns the number of seconds of CPU time used so far in the session. By saving the value of **time()** at the beginning of the execution group and subtracting that from the value at the end, we get the number of seconds spent doing the calculation.

```
> restart; timer:= time():
  int(sqrt(sec(x)^2+4), x = 0 .. 1);
  evalf(%);
  time()-timer;
```

$$\left(\left(-\frac{3}{125}+\frac{4}{125}\,\mathrm{I}\right)\sqrt{2}\left(10\,\mathrm{EllipticF}\left(\frac{1}{5}\,\frac{\sqrt{5}\,(-2-\mathrm{I}+2\cos(1)+\mathrm{I}\cos(1))}{\sin(1)},\frac{3}{5}-\frac{4}{5}\,\mathrm{I}\right)+5\,\mathrm{I}\,\mathrm{EllipticF}\left(\frac{1}{5}\,\frac{\sqrt{5}\,(-2-\mathrm{I}+2\cos(1)+\mathrm{I}\cos(1))}{\sin(1)},\frac{3}{5}-\frac{4}{5}\,\mathrm{I}\right)\right.\right.$$
$$-4\,\mathrm{EllipticPi}\left(\frac{\left(\frac{2}{5}+\frac{1}{5}\,\mathrm{I}\right)\sqrt{5}\,(-1+\cos(1))}{\sin(1)},\frac{3}{5}-\frac{4}{5}\,\mathrm{I},\frac{3}{5}-\frac{4}{5}\,\mathrm{I}\right)$$
$$-2\,\mathrm{I}\,\mathrm{EllipticPi}\left(\frac{\left(\frac{2}{5}+\frac{1}{5}\,\mathrm{I}\right)\sqrt{5}\,(-1+\cos(1))}{\sin(1)},\frac{3}{5}-\frac{4}{5}\,\mathrm{I},\frac{3}{5}-\frac{4}{5}\,\mathrm{I}\right)$$
$$-16\,\mathrm{EllipticPi}\left(\frac{\left(\frac{2}{5}+\frac{1}{5}\,\mathrm{I}\right)\sqrt{5}\,(-1+\cos(1))}{\sin(1)},-\frac{3}{5}+\frac{4}{5}\,\mathrm{I},\frac{3}{5}-\frac{4}{5}\,\mathrm{I}\right)$$
$$\left.-8\,\mathrm{I}\,\mathrm{EllipticPi}\left(\frac{\left(\frac{2}{5}+\frac{1}{5}\,\mathrm{I}\right)\sqrt{5}\,(-1+\cos(1))}{\sin(1)},-\frac{3}{5}+\frac{4}{5}\,\mathrm{I},\frac{3}{5}-\frac{4}{5}\,\mathrm{I}\right)\right)$$
$$\sin(1)^2\sqrt{40\cos(1)-20\,\mathrm{I}\cos(1)+10+20\,\mathrm{I}}\;\sqrt{4\cos(1)+2\,\mathrm{I}\cos(1)+1-2\,\mathrm{I}}$$

$$\left.\sqrt{1+4\cos(1)^2}\right) \Big/ \left((1+\cos(1))\left(4\cos(1)^3-4\cos(1)^2+\cos(1)-1\right)\right)$$

$$2.353900215-8.00000000\ 10^{-11}\ \mathrm{I}$$

$$3.713$$

```
> timer := time():
  evalf(Int(sqrt(sec(x)^2+4), x = 0 .. 1));
  time() - timer;
```

$$2.353900216$$

$$0.047$$

Of course, if your answer is not just a number (i.e. there are other parameters involved), **evalf(Int(...))** can't help.

```
> evalf(Int(sqrt(sec(x)^2+r^2), x = 0 .. 1)) assuming r > 0;
```

$$\int_{0.}^{1.}\sqrt{\sec(x)^2+r^2}\,\mathrm{d}x \tag{3.1}$$

Finding antiderivatives

How does Maple find formulas for antiderivatives?

In contrast to the completely mechanical procedure for differentiation, Calculus texts tend to have a hodgepodge of different techniques for integration, and often it is not clear which technique will work for a particular integral. If we can't manage to make these techniques work, there's always the suspicion that if we were cleverer we could find an answer. Maple uses a more systematic procedure that, in most cases, will produce an answer if there is one.

The basic problem can be stated as follows.

First, we need a technical term: **elementary function**. These include all the functions a Math 101 student might know about. Basically, we can start with constants and x, and build more and more complicated functions using:
• The arithmetic operations **+,-,*,/,^**
• Roots of polynomials whose coefficients are elementary functions
• **exp**, **ln**, and the standard trig and inverse trig functions

So for example $\sqrt{\sin(x)+\ln\left(1-\sqrt{x}\right)}$ is an elementary function, but $\mathrm{dilog}(x)$ and all sorts of other special functions are not elementary.

Now the question is this:
Given an elementary function $f(x)$, is there an elementary function $F(x)$ such that $F'(x)=f(x)$? If so, find it.

For example, for $f(x)=\sqrt{\tan(x)}$ the answer is yes, and here's one $F(x)$:

```
> int(sqrt(tan(x)),x);
```

$$\frac{1}{2}\,\frac{\sqrt{\tan(x)}\,\cos(x)\,\sqrt{2}\,\arccos(\cos(x)-\sin(x))}{\sqrt{\cos(x)\,\sin(x)}}-\frac{1}{2}\sqrt{2}\,\ln\left(\cos(x)\right) \tag{4.1}$$

$$+\sqrt{2}\,\sqrt{\tan(x)}\,\cos(x)+\sin(x)\Big)$$

On the other hand, for $f(x)=e^{x^2}$ the answer is no. Maple does find an antiderivative, but it's not an elementary function.

```
> int(exp(x^2),x);
```

$$-\frac{1}{2}\,\mathrm{I}\sqrt{\pi}\,\mathrm{erf}(\mathrm{I}\,x) \qquad \textbf{(4.2)}$$

There actually is an algorithm (called the **Risch algorithm**) that will answer this question. Some of the basic ingredients date back to Liouville in 1833, but the problem was not completely solved until quite recently: the last case of it only in 1987. The complete algorithm is not easy to implement, however, and Maple includes only part of it.

The theory involves some very high-powered mathematics, but I'll try to give you a taste of some of the simpler parts. We'll start by looking at how to integrate a rational function.

Integrating rational functions

A rational function is a quotient of polynomials. Here's a typical one.

```
> p := 2*x^5 + x^3 -x + 1:
  q := x^4 - 6*x^2 - 8*x - 3:
  f := p/q;
```

$$f:=\frac{2\,x^5+x^3-x+1}{x^4-6\,x^2-8\,x-3}$$

What is its antiderivative?

```
> F:=int(f,x);
```

$$F:=x^2+\frac{321}{64}\ln(x+1)+\frac{47}{16\,(x+1)}+\frac{511}{64}\ln(x-3)-\frac{1}{8\,(x+1)^2} \qquad \textbf{(5.1)}$$

How does the typical Calculus text tell you to integrate this (if it does)? The first step is division: write $\frac{p}{q}=s+\frac{r}{q}$ where s and r are polynomials, with $degree(r) < degree(q)$. That equation is equivalent to $p=s\,q+r$. Maple can do this with **quo** for quotient and **rem** for remainder, as we've seen.

```
> s := quo(p, q, x); r := rem(p, q, x);
```

$$s:=2\,x$$

$$r:=1+13\,x^3+5\,x+16\,x^2 \qquad \textbf{(5.2)}$$

```
> p/q = s + r/q;
```

$$\frac{2\,x^5+x^3-x+1}{x^4-6\,x^2-8\,x-3}=2\,x+\frac{1+13\,x^3+5\,x+16\,x^2}{x^4-6\,x^2-8\,x-3} \qquad \textbf{(5.3)}$$

Integrating the polynomial s is easy. That gives us the polynomial part of $F(x)$: x^2. We're left with the problem of integrating r/q, a rational function where the numerator has lower degree than the denominator.
The next step is to factor the denominator q.

```
> factor(q);
```

$$(x-3)\,(x+1)^3 \tag{5.4}$$

Now *r/q* is supposed to be decomposed into partial fractions: a sum of the following form:

```
> parfrac:= r/q = a/(x-3)+ b/(x+1) + c/(x+1)^2 + d/(x+1)^3;
```

$$parfrac := \frac{1+13\,x^3+5\,x+16\,x^2}{x^4-6\,x^2-8\,x-3} = \frac{a}{x-3} + \frac{b}{x+1} + \frac{c}{(x+1)^2} + \frac{d}{(x+1)^3}$$

To handle the $(x+1)^3$ we need a term in $(x+1)^{-j}$ for each *j* from 1 to 3.

We need to solve for the constants *a*, *b*, *c*, *d* that make this equation true. If we clear away the denominators, we get an equation for polynomials.

Maple objects introduced in this lesson

evalf(Int(...))
assume
about
assuming
time